Q:

Bug at meridian of a surface of revolution

 (* Dini (with Graphics error)*) 

 R = 2; 
 r = 1;

 X[u_] = R Sin[u];
 Z[u_] = -R (Cos[u] + Log[Tan[u/2]]); 

 ParametricPlot[{Z[u], X[u]}, {u, -Pi, Pi}, PlotLabel -> "PRE-TWIST MERIDIAN"] 

 h = 0; 
 ParametricPlot3D[{X[u] Cos[v], Z[u] + 0 h v, X[u] Sin[v]}, 
   {u, -Pi, Pi}, {v, 0, 3 Pi}, PlotLabel -> "MERIDIAN SWEPT AROUND Z-AXIS"]

 h = 1.5; 
 ParametricPlot3D[{X[u] Cos[v], Z[u] + h v,  X[u] Sin[v]}, 
  {u,-Pi,Pi}, {v, 0, 3 Pi}, Boxed -> False, Axes -> None, 
  PlotLabel ->"POST-TWIST ( h v) REVOLVED MERIDIAN"]

The knife edge graphics remanant shown below on untwisted meridian does not vanish unless the pseudosphere is twisted to a Dini.No such problems with PP3D in earlier Mathematica versions. 
EDIT1:

A:

Aside from increasing the plot points to PlotPoints -> 45, here's a way to delete the polygons that have long edges:
pl = ParametricPlot3D[{X[u] Cos[v], Z[u] + 0 h v, 
    X[u] Sin[v]}, {u, -Pi, Pi}, {v, 0, 3 Pi}, 
   PlotLabel -> "MERIDIAN SWEPT AROUND Z-AXIS"];
cliplen = 2;
With[{pts = pl[[1, 1]]},
 pl /. Polygon[polys_] :> Polygon@Pick[
     polys,
     Map[
      UnitStep[Max[EuclideanDistance @@@ Partition[pts[[#]], 2, 1, 1]] - cliplen] &,
      polys,
      {-2}],
     0]
 ]

ParametricPlot3D also does the correct meshing of the surface if the second coordinate is Z[u]/10 instead of Z[u] in the OP's code.
I'm not sure why what seem to possibly be scale issues (second coordinate rather large than the other two) would cause such trouble.  Setting Exclusions -> {Sin[u/2] == 0, Cos[u/2] == 0}, where the function has singularities, changes the graph but does not fix it.

